

ParaSwap AugustusV6 Security Audit Report

June 8, 2024

Contents

1	Intr	oduction	3
	1.1	About ParaSwap AugustusV6	3
	1.2	Audit Scope	3
	1.3	Changelog	4
2	Ove	erall Assessment	5
3	Vuli	nerability Summary	6
	3.1	Overview	6
	3.2	Security Level Reference	7
	3.3	Vulnerability Details	8
4	Арр	pendix	20
	4.1	About AstraSec	20
	4.2	Disclaimer	20
	13	Contact	21

1 Introduction

1.1 About ParaSwap AugustusV6

ParaSwap is a decentralized finance (DeFi) platform that aggregates liquidity from various decentralized exchanges (DEXs) to facilitate swift and efficient token swaps. It optimizes trades for the best available rates, minimizes slippage, and provides users with a user-friendly interface for seamless cryptocurrency trading across multiple blockchain networks. Compared with AugustusV5, the audited AugustusV6 introduces several key features:

- New gas optimised architecture for generic swap execution
- Gas efficient fee model
- Support of permit2
- ERC-2535 compatible Diamond proxy
- New FeeVault compatible with multiple versions of Augustus

1.2 Audit Scope

Augustus V6 Audit

The following source code was reviewed during the audit:

- https://github.com/paraswap/paraswap-contracts-v6/tree/fix/fixed-fees2
- CommitID: d405c95

And this is the final version representing all fixes implemented for the issues identified in the audit:

- https://github.com/paraswap/paraswap-augustus
- CommitID: eea7610

Note the src/executors/generic/Executor*.sol contracts are out of the audit scope.

Augustus V6.1 Audit

The following source code was reviewed during the audit:

- https://github.com/paraswap/paraswap-augustus/commits/deployment/v6.1/
- CommitID: 6093d46

And this is the final version representing all fixes implemented for the issues identified in the audit:

- https://github.com/paraswap/paraswap-augustus/commits/deployment/v6.1/
- CommitID: d45477e

Note the src/executors/generic/Executor*.sol contracts are out of the audit scope.

AugustusV6.2 Audit

The following source code was reviewed during the audit:

- https://github.com/paraswap/paraswap-augustus
- CommitID: d45477e

And this is the final version representing all fixes implemented for the issues identified in the audit:

- https://github.com/paraswap/paraswap-augustus
- CommitID: f000850

Note the src/executors/generic/Executor*.sol contracts are out of the audit scope.

1.3 Changelog

Version	Date
AugustusV6 Audit	January 9, 2024
AugustusV6.1 Audit	May 22, 2024
AugustusV6.2 Audit	May 27, 2024

2 Overall Assessment

This report has been compiled to identify issues and vulnerabilities within the ParaSwap AugustusV6 project. Throughout this audit, we identified several issues spanning various severity levels. By employing auxiliary tool techniques to supplement our thorough manual code review, we have discovered the following findings.

Severity	Count	Acknowledged	Won't Do	Addressed
Critical	-	-	-	-
High	3	-	-	3
Medium	4	-	-	4
Low	2	1	-	1
Informational	-	-	-	-
Undetermined	-	-	-	-

3 Vulnerability Summary

3.1 Overview

Click on an issue to jump to it, or scroll down to see them all.

- H-1 Incorrect Offset for Next Pool Address in UniswapV2Utils
- H-2 Revised Multi-Pool Swap in UniswapV3Utils
- H-3 Improper Logic with ETH as srcToken in swapExactAmountOutOnBalancerV2()
- M-1 Improper Calldata Overwritten in executeSwapOnCurveV2()
- M-2 Pulling Fee Failure under Permit2 Usage in AugustusFees
- M-3 Improper Parameters Order in swapExactAmountOutOnUniswapV3()
- M-4 Improper Fee Charge Logic in AugustusFees
- L-1 Improved swapType Decode for Curve Swap
- L-2 Potential Risks Associated with Centralization

3.2 Security Level Reference

In web3 smart contract audits, vulnerabilities are typically classified into different severity levels based on the potential impact they can have on the security and functionality of the contract. Here are the definitions for critical-severity, high-severity, medium-severity, and low-severity vulnerabilities:

Severity	Description
C-X (Critical)	A severe security flaw with immediate and significant negative consequences. It poses high risks, such as unauthorized access, financial losses,
	or complete disruption of functionality. Requires immediate attention and remediation.
H-X (High)	Significant security issues that can lead to substantial risks. Although
	not as severe as critical vulnerabilities, they can still result in unautho-
	rized access, manipulation of contract state, or financial losses. Prompt
	remediation is necessary.
M-X (Medium)	Moderately impactful security weaknesses that require attention and re-
	mediation. They may lead to limited unauthorized access, minor financial
	losses, or potential disruptions to functionality.
L-X (Low)	Minor security issues with limited impact. While they may not pose
	significant risks, it is still recommended to address them to maintain a
	robust and secure smart contract.
I-X (Informational)	Warnings and things to keep in mind when operating the protocol. No
	immediate action required.
U-X (Undetermined)	Identified security flaw requiring further investigation. Severity and im-
	pact need to be determined. Additional assessment and analysis are
	necessary.

3.3 Vulnerability Details

[H-1] Incorrect Offset for Next Pool Address in UniswapV2Utils

Target	Category	IMPACT	LIKELIHOOD	STATUS
UniswapV2Utils.sol	Business Logic	High	High	<i>⊗</i> Addressed

In the UniswapV2Utils contract, the _callUniswapV2PoolsSwapExactOut() function implements the exchange in UniswapV2 with an exact output amount. When it loads the next pool address from the memory, we notice it doesn't use the correct offset mul(add(i, 1), 32). Instead, it uses the offset mul(i, 32) (line 188) which is used to store the current pool address. As a result, the exchanges across multiple pools will fail.

```
UniswapV2Utils:: callUniswapV2PoolsSwapExactOut()
   // Loop for each pool
178 for { let i := 0 } lt(i, poolCount) { i := add(i, 1) } {
        // Check if it is the first pool
        if iszero(poolAddress) {...}
180
        // Adjust toAddress depending on if it is the last pool in the array
182
        let toAddress := address()
183
        // Check if it is not the last pool
        if lt(add(i, 1), poolCount) {
186
            // Load next pool address
187
            nextPoolAddress := mload(add(poolAddresses, mul(i, 32)))
            // Adjust toAddress to next pool address
190
            toAddress := nextPoolAddress
        }
192
193
194 }
```

Remediation Use the correct offset mul(add(i, 1), 32) to load the next pool address.

[H-2] Revised Multi-Pool Swap in UniswapV3Utils

Target	Category	IMPACT	LIKELIHOOD	STATUS
UniswapV3Utils.sol	Business Logic	High	High	<i>⊗</i> Addressed

In the UniswapV3Utils contract, the UniswapV3Utils::_callUniswapV3PoolsSwap() routine implements the exchange in UniswapV3 with an exact input amount. For the exchange across multiple

pools, this contract, i.e., adderss(this), shall serve as both the input payer and the output recipient. However, we notice that the current implementation uses the next pool as the recipient for the current exchange (line 338). This will lead to the exchange failure in the next pool, because this contract, i.e., the payer, has no fund to pay the exchange.

```
UniswapV3Utils:: callUniswapV3PoolsSwap()
   if lt(add(i, 1), poolCount) {
307
308
       // Calculate Next Pool Address
310
       // Store Oxff + factory address (right padded)
311
       mstore(ptr, uniswapV3FactoryAndFF)
       // Store pools offset + 21 bytes (UNISWAP_V3_FACTORY_AND_FF SIZE)
314
       let tokenOptr := add(ptr, 21)
317
       // Copy next pool data to free memory pointer + 21 bytes (
           UNISWAP_V3_FACTORY_AND_FF SIZE)
318
       calldatacopy(add(token0ptr, 1), add(add(pools.offset, 1), mul(add(i, 1), 96)
       // Calculate keccak256(abi.encode(address(token0), address(token1), fee))
320
       mstore(tokenOptr, keccak256(tokenOptr, 96))
321
       // Store POOL_INIT_CODE_HASH
323
       mstore(add(token0ptr, 32), uniswapV3PoolInitCodeHash)
       // Calculate keccak256(abi.encodePacked(hex'ff', address(factory_address),
326
       // keccak256(abi.encode(token0,
327
       // token1, fee)), POOL_INIT_CODE_HASH));
328
       mstore(ptr, keccak256(ptr, 85)) // 21 + 32 + 32
329
       // Load pool
331
       p := mload(ptr)
332
       // Get the first 20 bytes of the computed address
334
       335
       // Adjust toAddress to next pool address
337
       toAddress := nextPoolAddress
338
339 }
```

Moreover, the UniswapV3Utils::_callUniswapV3PoolsSwapExactAmountOut() routine implements the exchange in UniswapV3 with an exact output amount. Our analysis shows that it doesn't properly implement the exchange across multiple pools.

To support the exchanges across multiple pools, it is necessary to call the pool.swap() function recursively, with the output of each step serving as the input for the previous step, thus completing

the entire exchange process. We suggest to refer to the SwapRouter contract from Uniswap for implementing exchanges across multiple pools.

Remediation Refer to the SwapRouter contract from Uniswap for implementing exchanges across multiple pools.

[H-3] Improper Logic with ETH as srcToken in swapExactAmountOutOnBalancerV2()

Target	Category	IMPACT	LIKELIHOOD	STATUS
BalancerV2SwapExactAmountOut.sol	Business Logic	High	Medium	<i>⊗</i> Addressed
${\sf GenericSwapExactAmountOut.sol}$				

The swapExactAmountOutOnBalancerV2() function is designed to swap srcToken for a specified amount of destToken via BalancerV2. Within this function, it initially obtain the amount of srcToken currently held by the contract before executing the swap (line 45). This is done to calculate the remaining amount of srcToken after the swap. Upon thorough examination, we observed that if srcToken is ETH, the amount obtained before the swap includes the incoming msg.value, which is clearly not intended. This leads to an incorrect calculation of the remaining amount (line 56).

```
BalancerV2SwapExactAmountOut::swapExactAmountOutOnBalancerV2()
31 function swapExactAmountOutOnBalancerV2(
       BalancerV2Data calldata balancerData,
32
33
       uint256 partnerAndFee,
       bytes calldata permit,
34
       bytes calldata data
35
36 )
37
       external
       payable
38
       whenNotPaused
       returns (uint256 spentAmount, uint256 receivedAmount, uint256 paraswapShare,
40
             uint256 partnerShare)
41 {
42
       . . .
       // Check contract balance
       uint256 balanceBefore = srcToken.getBalance(address(this));
45
47
        . . .
49
       // Execute swap
50
       _callBalancerV2(data);
       // Check balance of destToken
```

```
receivedAmount = destToken.getBalance(address(this));
53
       // Check balance of srcToken, deducting the balance before the swap if it is
            greater than 1
       uint256 remainingAmount = srcToken.getBalance(address(this)) - (
56
           balanceBefore > 1 ? balanceBefore : 0);
       // Check if swap succeeded
58
       if (receivedAmount < amountOut) {</pre>
           revert InsufficientReturnAmount();
60
61
       // Process fees and transfer destToken and srcToken to beneficiary
63
       return processSwapExactAmountOutFeesAndTransfer(
65
           beneficiary,
           srcToken,
66
           destToken,
           partnerAndFee,
68
69
           maxAmountIn,
           remainingAmount,
71
           receivedAmount,
           quotedAmountIn
72
73
       );
74 }
```

Remediation Properly handle the incoming msg.value when srcToken is ETH in the swapExactAmount -OutOnBalancerV2() and swapExactAmountOut() functions.

[M-1] Improper Calldata Overwritten in _executeSwapOnCurveV2()

Target	Category	IMPACT	LIKELIHOOD	STATUS
CurveV2SwapExactAmountIn.	sœBusiness Logic	Medium	Medium	<i>⊙</i> Addressed

The _executeSwapOnCurveV2() function in the CurveV2SwapExactAmountIn contract implements the exchanges in CurveV2. Before invoking the exchange() function of CurveV2, it assembles the calldata according to the input data, e.g., poolAddress, i, j, fromAmount.

However, we notice the assembled calldata is erroneously overwritten, potentially leading to a failed call to the exchange (line 212).

```
mstore(ptr, 0
206
           mstore(add(ptr, 4), poolAddress) // store poolAddress
       mstore(add(ptr, 36), i) // store index i
208
209
       mstore(add(ptr, 68), j) // store index j
210
       mstore(add(ptr, 100), fromAmount) // store fromAmount
       mstore(add(ptr, 132), 1) // store 1
211
       calldatacopy(add(ptr, 4), 164, 160)
212
213
       // Perform the external call with the prepared calldata
       // Check the outcome of the call and handle failure
214
       if iszero(call(gas(), exchange, 0, ptr, 164, 0, 0)) {
215
          // The call failed; we retrieve the exact error message and revert with
216
              it
           returndatacopy(0, 0, returndatasize()) // Copy the error message to the
217
              start of memory
          revert(0, returndatasize()) // Revert with the error message
218
       }
219
220 }
```

Remediation Properly remove the overwrite of the assembled calldata.

[M-2] Pulling Fee Failure under Permit2 Usage in AugustusFees

Target	Category	IMPACT	LIKELIHOOD	STATUS
UniswapV3SwapExactAmountOut.sol	Business Logic	Medium	Medium	<i>⊗</i> Addressed
AugustusFees.sol				

The swapExactAmountOutOnUniswapV3() function is employed to exchange a specified quantity of a given token via UniswapV3. It allows users to grant approval to the UniswapV3SwapExactAmountOut contract via permit2 (line 53). Inside it, the processSwapExactAmountOutFeesAndTransferUniV3() function is invoked (line 65) after the swap in order to charge fee and transfer the destination token to the beneficiary. Ultimately, the _distributeFeesUniV3() function is called to handle fee distribution. Upon thorough examination, it comes to our attention that this function does not consider scenarios where users grant approval through permit2. Consequently, this oversight may result in fee distribution failures in such cases.

```
UniswapV3SwapExactAmountOut::swapExactAmountOutOnUniswapV3()

40 function swapExactAmountOutOnUniswapV3(
41 UniswapV3Data calldata uniData, uint256 partnerAndFee, bytes calldata permit
42 ) external payable returns (uint256 receivedAmount, uint256 spentAmount) {
43 ...
44 // Check if we need to wrap or permit
```

```
if (isFromETH) {
45
46
            . . .
       } else {
47
48
            if (permit.length < 257) {</pre>
49
           } else {
51
                // Otherwise Permit2.permit
52
53
                permit2Approve(permit);
                permit2 = 1;
54
           }
55
            // Swap will be paid from msg.sender
56
           fromAddress = msg.sender;
57
       }
58
59
       . . .
       if (isFromETH) {
60
       } else {
62
           // Process fees and transfer destToken and srcToken to feeVault or
63
                partner and
            // feeWallet if needed
64
            receivedAmount = processSwapExactAmountOutFeesAndTransferUniV3(
65
                beneficiary, srcToken, destToken, partnerAndFee, maxAmountIn,
67
                receivedAmount, spentAmount, quotedAmountIn
           );
68
       }
69
70 }
```

AugustusFees::processSwapExactAmountOutFeesAndTransferUniV3()& distributeFeesUniV3()

```
{\tt 39} \quad \textbf{function} \quad \texttt{processSwapExactAmountOutFeesAndTransferUniV3} \, (
        address beneficiary, IERC20 srcToken, IERC20 destToken,
       uint256 partnerAndFee, uint256 fromAmount, uint256 receivedAmount,
       uint256 spentAmount, uint256 quotedAmount
42
43 ) internal returns (uint256 returnAmount) {
44
       if (partner != address(0x0)) {
45
46
            if (feeData & IS_REFERRAL_MASK != 0) {
47
                if (surplus > 0) {
48
                    // the split is 50% for paraswap, 25% for the referrer and 25%
                         for the user
                    uint256 paraswapShare = (surplus * PARASWAP_REFERRAL_SHARE) / 10
50
                         _000;
                    uint256 referrerShare = (paraswapShare * 5000) / 10_000;
51
52
                    // distribute fees from srcToken
53
                     _distributeFeesUniV3(
                         remainingAmount, msg.sender, srcToken, partner,
54
                         referrerShare, paraswapShare,
55
```

```
skipWhitelist // Check if skipWhitelist flag is true
56
                    );
57
                    return (receivedAmount);
                }
59
           }
60
       }
62
63
64 }
66 function _distributeFeesUniV3(
       uint256 currentBalance, address payer, IERC20 token,
67
       address payable partner, uint 256 partner Share, uint 256 paraswap Share,
68
       bool skipWhitelist
69
   ) private {
70
       uint256 totalFees = partnerShare + paraswapShare;
71
       if (totalFees != 0) {
            if (totalFees > currentBalance) {
                revert InsufficientBalanceToPayFees();
74
           }
           if (skipWhitelist) {
76
                \ensuremath{//} transfer the fees directly to the partner and feeWallet
77
78
                if (paraswapShare > 0) {
                    token.safeTransferFrom(payer, feeWallet, paraswapShare);
79
                }
80
                if (partnerShare > 0) {
                    token.safeTransferFrom(payer, partner, partnerShare);
82
                }
83
           } else {
84
                // transfer the fees to the fee vault
85
                token.safeTransferFrom(payer, address(FEE_VAULT), totalFees);
86
87
                if (paraswapShare > 0) {
                    FEE_VAULT.registerFee(feeWalletDelegate, token, paraswapShare);
88
                }
89
                if (partnerShare > 0) {
90
                    FEE_VAULT.registerFee(partner, token, partnerShare);
91
92
93
94
            \ensuremath{//} othwerwise do not transfer the fees
95
       }
96 }
```

Remediation Consider situations where users grant approval through permit2.

[M-3] Improper Parameters Order in swapExactAmountOutOnUniswapV3()

Target	Category	IMPACT	LIKELIHOOD	STATUS
Uniswap V3 Swap Exact Amount Out. sol	Business Logic	Medium	Medium	<i>⊗</i> Addressed

As previously mentioned, the swapExactAmountOutOnUniswapV3() function is employed to exchange a specified quantity of a given token via UniswapV3. Within it, if the source token is ETH, the processSwapExactAmountOutFeesAndTransfer() function is invoked (line 54) after the swap in order to handle fee distribution and transfer both the destination token and any remaining source token to the beneficiary. Upon further analysis, we noticed that the sixth parameter (i.e., receivedAmount, line 60) and the seventh parameter (i.e., remainingAmount, line 61) are set in the wrong order, which may lead to unexpected or undesired outcomes.

```
UniswapV3SwapExactAmountOut::swapExactAmountOutOnUniswapV3()
40 function swapExactAmountOutOnUniswapV3(
       UniswapV3Data calldata uniData, uint256 partnerAndFee, bytes calldata permit
42 ) external payable returns (uint256 receivedAmount, uint256 spentAmount) {
       // Check if srcToken is ETH and unwrap
       if (isFromETH) {
45
           uint256 remainingAmount = maxAmountIn - spentAmount;
           // Withdraw remaining WETH if any
48
           if (remainingAmount > 0) {
49
               WETH.withdraw(remainingAmount);
51
           // Process fees using processSwapExactAmountOutFeesAndTransfer
53
           (spentAmount, receivedAmount) = processSwapExactAmountOutFeesAndTransfer
54
               beneficiary,
55
               ERC20Utils.ETH,
56
               destToken,
               partnerAndFee,
58
59
               maxAmountIn,
               receivedAmount,
               remainingAmount,
61
                quotedAmountIn
62
           );
63
       } else {
64
65
           . . .
       }
66
67 }
```

AugustusFees::processSwapExactAmountOutFeesAndTransfer() 203 function processSwapExactAmountOutFeesAndTransfer(address beneficiary, 204 205 IERC20 srcToken, IERC20 destToken, 206 uint256 partnerAndFee, 207 uint256 fromAmount, 208 209 uint256 remainingAmount, uint256 receivedAmount, 210 uint256 quotedAmount 212) internal returns (uint256 spentAmount, uint256 returnAmount)

Remediation Rectify the parameter order to ensure the function behaves as intended.

[M-4] Improper Fee Charge Logic in AugustusFees

Target	Category	IMPACT	LIKELIHOOD	STATUS
AugustusFees.sol	Business Logic	Medium	Low	<i>⊗</i> Addressed

The AugustusFees contract implements the logic for charging fees after the completion of swaps, and sends the received/remaining tokens to the beneficiary. While reviewing its logic, we notice the current fee charging logic needs to be revisited.

To elaborate, we take the processSwapExactAmountInFeesAndTransfer() routine for example and show blow the code snippet from the AugustusFees. The fee logic that the contract aims to implement is as follows: if the IS_REFERRAL_MASK flag is set to true and surplus > 0, then the fee will be charged from the surplus. Otherwise, if the IS_TAKE_SURPLUS_MASK flag is set to true and surplus > 0, then the fee will be charged from the surplus. If neither the IS_REFERRAL_MASK nor IS_TAKE_SURPLUS_MASK flags are set, then fees will be charged based on the feePercent configured in the function input parameters. However, the current code implementation allows the fee logic to be bypassed if surplus == 0 and either the IS_REFERRAL_MASK or IS_TAKE_SURPLUS_MASK flag is set to true.

```
AugustusFees::processSwapExactAmountInFeesAndTransfer()

function processSwapExactAmountInFeesAndTransfer(

address beneficiary,

IERC20 destToken,

uint256 partnerAndFee,

uint256 receivedAmount,

uint256 quotedAmount

internal
```

```
returns (uint256 returnAmount)
82
   {
83
84
        // if partner address is not 0x0
85
        if (partner != address(0x0)) {
86
87
            // if slippage is postive and referral flag is true
88
            if (feeData & IS_REFERRAL_MASK != 0) {
89
90
                if (surplus > 0) {
91
                     . . .
                }
92
            }
93
            // if slippage is positive and takeSurplus flag is true
94
            else if (feeData & IS_TAKE_SURPLUS_MASK != 0) {
                if (surplus > 0) {
96
97
                     . . .
                }
99
            }
            // partner takes fixed fees if isTakeSurplus and isReferral flags are
100
            // and feePercent is greater than 0
101
            else if (feeData & IS_TAKE_SURPLUS_MASK == 0 && feeData &
102
                IS_REFERRAL_MASK == 0) {
                uint256 feePercent = _getAdjustedFeePercent(feeData);
103
                if (feePercent > 0) {
104
105
                }
106
            }
107
        }
108
110
111 }
```

Note the processSwapExactAmountOutFeesAndTransfer() and processSwapExactAmountOutFeesAndTransferUniV3 () routines share the same issue.

Remediation Implement the correct fee charging logic when surplus == 0.

[L-1] Improved swapType Decode for Curve Swap

Target	Category	IMPACT	LIKELIHOOD	STATUS
CurveV1SwapExactAmountIn.sol	Coding Practices	Low	Low	<i>⊗</i> Addressed
CurveV2SwapExactAmountIn.sol				

In the CurveV1SwapExactAmountIn contract, the swapExactAmountInOnCurveV1() function is implemented to execute a swapExactAmountIn on Curve V1 pools. While examining its logic, we notice that the use of one input parameter is inconsistent with the descriptions provided in the comments for

this parameter.

To elaborate, we show below the related code snippet of the CurveV1SwapExactAmountIn contract. Based on the comments, the function input parameter curveV1Data is encapsulated as follows: the first 160 bits is the target exchange address, bit 161 is the approve flag, bits 162 - 163 are used for the wrap flag, bits 164 - 165 are used for the swapType flag, and the last 91 bits are unused. However, in the current code implementation, bits 165-166 are used as the swapType flag, instead of bits 164 - 165. This may bring unnecessary hurdles to understand and/or maintain the smart contract.

```
CurveV1SwapExactAmountIn::swapExactAmountInOnCurveV1()
{\tt 33} \quad \textbf{function} \quad {\tt swapExactAmountInOnCurveV1} \, (
       CurveV1Data calldata curveV1Data,
      uint256 partnerAndFee,
35
      bytes calldata permit
36
37 )
       external
38
40
       returns (uint256 receivedAmount)
41 {
42
      // solhint-disable-next-line no-inline-assembly
44
       assembly ("memory-safe") {
45
          46
          approveFlag := and(shr(160, curveData), 1)
47
          wrapFlag := and(shr(161, curveData), 3)
48
          swapType := and(shr(164, curveData), 3)
49
      }
50
52
53 }
```

Note that the CurveV2SwapExactAmountIn contract shares the same issue.

Remediation Ensure the consistency between documents and implementation.

[L-2] Potential Risks Associated with Centralization

Target	Category	IMPACT	LIKELIHOOD	STATUS
Multiple Contracts	Security	Low	Low	Acknowledged

In the ParaSwap AugustusV6 protocol, the existence of a privileged owner account introduces centralization risks, as it holds significant control and authority over critical operations governing the protocol. In the following, we show the representative function potentially affected by the privileges associated with the privileged account.

```
Example Privileged Operations in ParaSwap AugustusV6
27 function executeSelfdestruct() external onlyOwner {
       // Selfdestruct the contract
       selfdestruct(payable(msg.sender));
29
30 }
32 function setfeeWalletDelegate(address payable _feeWalletDelegate) external
       onlyOwner {
       // Make sure the fee wallet is not the zero address
33
      if (_feeWalletDelegate == address(0)) revert InvalidWalletAddress();
34
       // Set the fee wallet
36
       feeWalletDelegate = _feeWalletDelegate;
37 }
39 function setTokenBlacklisting(IERC20 token, bool isBlacklisted) public onlyOwner
       // Set the blacklisting status
       blacklistedTokens[token] = isBlacklisted;
41
42
       // Emit an event
       emit TokenBlacklistUpdated(token, isBlacklisted);
44 }
```

Remediation To mitigate the identified issue, it is recommended to introduce multi-sig mechanism to undertake the role of the privileged account. Moreover, it is advisable to implement timelocks to govern all modifications to the privileged operations.

4 Appendix

4.1 About AstraSec

AstraSec is a blockchain security company that serves to provide high-quality auditing services for blockchain-based protocols. With a team of blockchain specialists, AstraSec maintains a strong commitment to excellence and client satisfaction. The audit team members have extensive audit experience for various famous DeFi projects. AstraSec's comprehensive approach and deep blockchain understanding make it a trusted partner for the clients.

4.2 Disclaimer

The information provided in this audit report is for reference only and does not constitute any legal, financial, or investment advice. Any views, suggestions, or conclusions in the audit report are based on the limited information and conditions obtained during the audit process and may be subject to unknown risks and uncertainties. While we make every effort to ensure the accuracy and completeness of the audit report, we are not responsible for any errors or omissions in the report.

We recommend users to carefully consider the information in the audit report based on their own independent judgment and professional advice before making any decisions. We are not responsible for the consequences of the use of the audit report, including but not limited to any losses or damages resulting from reliance on the audit report.

This audit report is for reference only and should not be considered a substitute for legal documents or contracts.

4.3 Contact

Name	AstraSec Team
Phone	+86 176 2267 4194
Email	contact@astrasec.ai
Twitter	https://twitter.com/AstraSecAI