



Eigenpie

Security Audit Report

May 4, 2024

---

# Contents

<b>1</b>	<b>Introduction</b>	<b>3</b>
1.1	About Eigenpie . . . . .	3
1.2	Audit Scope . . . . .	3
1.3	Changelog . . . . .	4
<b>2</b>	<b>Overall Assessment</b>	<b>5</b>
<b>3</b>	<b>Vulnerability Summary</b>	<b>6</b>
3.1	Overview . . . . .	6
3.2	Security Level Reference . . . . .	7
3.3	Vulnerability Details . . . . .	8
<b>4</b>	<b>Appendix</b>	<b>19</b>
4.1	About AstraSec . . . . .	19
4.2	Disclaimer . . . . .	19
4.3	Contact . . . . .	20

---

# 1 | Introduction

## 1.1 About Eigenpie

Eigenpie is a re-staking platform for SubDAO, providing Liquid Stake Token (LST) holders with the ability to re-stake their assets and maximize their profit potential. It achieves this by creating dedicated liquidity restaking for each accepted LST on its platform, effectively isolating risks associated with any particular LST.

## 1.2 Audit Scope

### First Audit Scope

The following source code was reviewed during the audit:

- <https://github.com/magpiexyz/eigenpie.git>
- Commit ID: 297d1ba

And this is the final version representing all fixes implemented for the issues identified in the audit:

- <https://github.com/magpiexyz/eigenpie.git>
- Commit ID: 72227d5

### Second Audit Scope

The following source code was reviewed during the audit:

- <https://github.com/magpiexyz/eigenpie/commits/feature/withdrawStakedETH>
- Commit ID: eda5ddb

And this is the final version representing all fixes implemented for the issues identified in the audit:

- <https://github.com/magpiexyz/eigenpie/commits/feature/withdrawStakedETH>
- Commit ID: 57d825e

---

## Third Audit Scope

The following source code was reviewed during the audit:

- <https://github.com/magpiexyz/eigenpie/pull/19/commits>
- Commit ID: 7210994

And this is the final version representing all fixes implemented for the issues identified in the audit:

- <https://github.com/magpiexyz/eigenpie/pull/19/commits>
- Commit ID: f2bd5d0

## 1.3 Changelog

Version	Date
First Audit	February 14, 2024
Second Audit	March 23, 2024
Third Audit	May 3, 2024

---

## 2 | Overall Assessment

This report has been compiled to identify issues and vulnerabilities within the `Eigenpie` project. Throughout this audit, we identified several issues spanning various severity levels. By employing auxiliary tool techniques to supplement our thorough manual code review, we have discovered the following findings.

Severity	Count	Acknowledged	Won't Do	Addressed
Critical	-	-	-	-
High	6	-	-	6
Medium	2	1	-	1
Low	2	-	-	2
Informational	-	-	-	-
Undetermined	-	-	-	-

---

## 3 | Vulnerability Summary

### 3.1 Overview

Click on an issue to jump to it, or scroll down to see them all.

- [H-1](#) Revised Pre-Deposit Logic in `EigenpieStaking::depositAsset()`
- [H-2](#) Improper exchangeRate Precision in `PriceProvider::updateMLRTPrice(address)`
- [H-3](#) Improper Implementation of `PriceProvider::updateMLRTPrice(address, uint256)`
- [H-4](#) Revised CleanUp of Withdrawal Schedules
- [H-5](#) Reserve ETH for New Validators in `NodeDelegator::receive()`
- [H-6](#) Flawed Exchange Rate Calculation
- [M-1](#) Potential Risks Associated with Centralization
- [M-2](#) Revised `assetTotalWithdrawAmt` Check
- [L-1](#) Integration of Non-Standard ERC20 Tokens
- [L-2](#) Improved Logic of `MLRTWallet::withdrawFromSwellStaking()`

---

## 3.2 Security Level Reference

In web3 smart contract audits, vulnerabilities are typically classified into different severity levels based on the potential impact they can have on the security and functionality of the contract. Here are the definitions for critical-severity, high-severity, medium-severity, and low-severity vulnerabilities:

Severity	Description
C-X (Critical)	A severe security flaw with immediate and significant negative consequences. It poses high risks, such as unauthorized access, financial losses, or complete disruption of functionality. Requires immediate attention and remediation.
H-X (High)	Significant security issues that can lead to substantial risks. Although not as severe as critical vulnerabilities, they can still result in unauthorized access, manipulation of contract state, or financial losses. Prompt remediation is necessary.
M-X (Medium)	Moderately impactful security weaknesses that require attention and remediation. They may lead to limited unauthorized access, minor financial losses, or potential disruptions to functionality.
L-X (Low)	Minor security issues with limited impact. While they may not pose significant risks, it is still recommended to address them to maintain a robust and secure smart contract.
I-X (Informational)	Warnings and things to keep in mind when operating the protocol. No immediate action required.
U-X (Undetermined)	Identified security flaw requiring further investigation. Severity and impact need to be determined. Additional assessment and analysis are necessary.

### 3.3 Vulnerability Details

#### [H-1] Revised Pre-Deposit Logic in EigenpieStaking::depositAsset()

Target	Category	IMPACT	LIKELIHOOD	STATUS
EigenpieStaking.sol EigenpiePreDepositHelper.sol	Business Logic	High	High	<a href="#">Addressed</a>

The `EigenpieStaking::depositAsset()` function serves as a mechanism for users to deposit supported LST (e.g., `ankrETH`, `cbETH`, etc.) and the corresponding `mLRT-LST` token is minted. During the pre-deposit phase of the protocol, users deposit underlying token into the `EigenpieStaking` contract (line 177). However, the corresponding `mLRT-LST` token is not immediately minted for them (lines 166 - 168). Instead, they need to wait until the current pre-deposit cycle is concluded. Upon claiming (line 84), the `mLRT-LST` token will then be minted and allocated to the users (line 93). This may result in the `totalSupply` of the `mLRT-LST` token is not updated in time, which is crucial for calculating the `mLRT-LST/LST` exchange rate. Consequently, it will lead to inaccuracy in the exchange rate calculation.

#### EigenpieStaking::depositAsset()

```
145 function depositAsset(  
146     address asset,  
147     uint256 depositAmount,  
148     uint256 minRec,  
149     address referral  
150 )  
151     external  
152     whenNotPaused  
153     nonReentrant  
154     onlySupportedAsset(asset)  
155 {  
156     // checks  
157     if (depositAmount == 0 || depositAmount < minAmountToDeposit) {  
158         revert InvalidAmountToDeposit();  
159     }  
  
161     if (depositAmount > getAssetCurrentLimit(asset)) {  
162         revert MaximumDepositLimitReached();  
163     }  
  
165     uint256 mintedAmount;  
166     if (isPreDeposit) {  
167         (mintedAmount,) = getMLRTAmountToMint(asset, depositAmount);  
168         IEigenpiePreDepositHelper(eigenpiePreDepositHelper).feedUserDeposit(msg.  
169             sender, asset, mintedAmount);  
169     } else {
```



```

170         // mint receipt
171         mintedAmount = _mintMLRT(asset, depositAmount);
172     }
173     if (mintedAmount < minRec) {
174         revert MinimumAmountToReceiveNotMet();
175     }

177     IERC20(asset).safeTransferFrom(msg.sender, address(this), depositAmount);

179     emit AssetDeposit(msg.sender, asset, depositAmount, referral);
180 }

```

#### EigenpiePreDepositHelper::userClaim()

```

84 function userClaim(uint256[] calldata _cycles, address[] calldata _assets)
    external nonReentrant {
85     for (uint256 i = 0; i < _cycles.length; i++) {
86         if (!claimableCycles[_cycles[i]]) revert ClaimCycleNotStarted();
87         for (uint256 j = 0; j < _assets.length; j++) {
88             bytes32 cycleUserKey = this._getCycleUserKey(_cycles[i], msg.sender)
                ;
89             UserInfo storage user = userInfo[cycleUserKey][_assets[j]];
90             uint256 amount = user.amount - user.claimed;
91             if (amount > 0) {
92                 address receipt = eigenpieConfig.mLRTReceiptByAsset(_assets[j]);
93                 IMintableERC20(receipt).mint(msg.sender, amount);
94                 user.claimed += amount;
95                 emit Claim(msg.sender, _assets[j], amount, _cycles[i]);
96             }
97         }
98     }
99 }

```

**Remediation** Ensure the totalSupply of the mLRT-LST token is updated in time.

## [H-2] Improper exchangeRate Precision in PriceProvider::updateMLRTPrice(address)

Target	Category	IMPACT	LIKELIHOOD	STATUS
PriceProvider.sol	Business Logic	High	High	<a href="#">Addressed</a>

The PriceProvider::updateMLRTPrice(address) function is utilized to update the mLRT-LST/LST exchange rate for the specified asset. The exchange rate is derived from the current state of the corresponding pool. During our examination of the exchange rate calculation logic, it is apparent

that there is a loss of precision for the result. Given this, we suggest to improve its implementation as below: `uint256` `exchangeRate = totalLST * 1 ether / receiptSupply` (line 69).

Moreover, to mitigate potential front-run attacks, we recommend adding access control to this function and execute transactions for updating the exchange rate through private RPC (e.g., `flashbot`).

#### PriceProvider::updateMLRTPrice(address)

```
54 /// @notice updates mLRT-LST/LST exchange rate
55 /// @dev calculates based on stakedAsset value received from eigen layer
56 /// @param asset the asset for which exchange rate to update
57 function updateMLRTPrice(address asset) external {
58     address mLRTReceipt = eigenpieConfig.mLRTReceiptByAsset(asset);
59     uint256 receiptSupply = IMLRT(mLRTReceipt).totalSupply();

61     if (receiptSupply == 0) {
62         IMLRT(mLRTReceipt).updateExchangeRateToLST(1 ether);
63         return;
64     }

66     address eigenStakingAddr = eigenpieConfig.getContract(EigenpieConstants.
        EIGENPIE_STAKING);
67     uint256 totalLST = IEigenpieStaking(eigenStakingAddr).getTotalAssetDeposits(
        asset);

69     uint256 exchangeRate = totalLST / receiptSupply;

71     _checkNewRate(mLRTReceipt, exchangeRate);

73     IMLRT(mLRTReceipt).updateExchangeRateToLST(exchangeRate);
74 }
```

**Remediation** Correct the implementation of the `PriceProvider::updateMLRTPrice(address)` function as above mentioned.

### [H-3] Improper Implementation of `PriceProvider::updateMLRTPrice(address, uint256)`

Target	Category	IMPACT	LIKELIHOOD	STATUS
PriceProvider.sol	Business Logic	High	High	<a href="#">Addressed</a>

As part of its intended functionality, the `PriceProvider::updateMLRTPrice(address, uint256)` function is employed by the privileged account to manually adjust the exchange rate based on off-chain calculations, thereby optimizing gas usage. However, thorough examination of its implementation,

we observed that it lacks any form of access control and does not actually modify the exchange rate, which clearly deviates from the intended design.

#### PriceProvider::updateMLRTPrice(address, uint256)

```
76 /// @notice updates mLRT-LST/LST exchange rate manually for gas fee saving
77 /// @dev calculates based on stakedAsset value received from eigen layer
78 /// @param asset the asset for which exchange rate to update
79 /// @param newExchangeRate the new exchange rate to update
80 function updateMLRTPrice(address asset, uint256 newExchangeRate) external {
81     address mLRTReceipt = eigenpieConfig.mLRTReceiptByAsset(asset);

83     _checkNewRate(mLRTReceipt, newExchangeRate);

85     emit ExchangeRateUpdate(asset, mLRTReceipt, newExchangeRate);
86 }
```

**Remediation** Apply necessary access control and properly update the exchange rate.

## [H-4] Revised CleanUp of Withdrawal Schedules

Target	Category	IMPACT	LIKELIHOOD	STATUS
EigenpieWithdrawManager.sol	Business Logic	High	High	<a href="#">Addressed</a>

The EigenpieWithdrawManager contract exclusively manages the queuing and withdrawal processes of tokens, specifically LSTs (Liquid Staking Tokens). Users queue for the withdrawals of specific assets and proceed with the withdrawal after the withdrawal period expires. Then the withdrawals are removed from the schedules list.

While reviewing the withdraw logic, we notice that the claimed withdrawals may not be correctly removed from the schedules list. There are two key points of concern. Firstly, the `userWithdrawAsset()` function does not accurately count the total number of claimed withdrawals that can be removed (i.e., `claimedWithdrawalSchedulesPerAsset`), but only counts the number of new claimed withdrawals in the current transaction (line 157). This leads to the `_cleanUpWithdrawalSchedules()` function being unable to correctly determine whether to proceed with the removal based on the threshold check (`withdrawalscheduleCleanUp`).

#### EigenpieWithdrawManager::userWithdrawAsset()

```
139 function userWithdrawAsset(address[] memory assets) external nonReentrant {
140     uint256[] memory claimedWithdrawalSchedules = new uint256[](assets.length);

142     for (uint256 i = 0; i < assets.length; i++) {
```

```

143     bytes32 userToAsset = userToAssetKey(msg.sender, assets[i]);
144     WithdrawalSchedule[] storage schedules = withdrawalSchedules[userToAsset];

146     uint256 totalClaimedAmount;
147     uint256 burnAmount;
148     uint256 claimedWithdrawalSchedulesPerAsset;

150     for (uint256 j = 0; j < schedules.length; j++) {
151         WithdrawalSchedule storage schedule = schedules[j];
152         if (block.timestamp >= schedule.endTime) {
153             uint256 availableToClaim = schedule.queuedWithdrawLSTAmt;

155             if (availableToClaim >= schedule.claimedAmt) {
156                 ...
157                 claimedWithdrawalSchedulesPerAsset++;
158             }
159         }
160     }

162     claimedWithdrawalSchedules[i] = claimedWithdrawalSchedulesPerAsset;

164     if (totalClaimedAmount > 0) {...}
165 }

167 _cleanUpWithdrawalSchedules(assets, claimedWithdrawalSchedules);
168 }

```

Secondly, a redundant for-loop (line 272) is added in `_cleanUpWithdrawalSchedules()` which brings the possibility of claimed withdrawals being removed from the schedules list repeatedly.

#### EigenpieWithdrawManager::\_cleanUpWithdrawalSchedules()

```

262 function _cleanUpWithdrawalSchedules(
263     address[] memory assets,
264     uint256[] memory clamiedWithdrawalSchedules
265 )
266     internal
267 {
268     for (uint256 i = 0; i < assets.length; i++) {
269         bytes32 userToAsset = userToAssetKey(msg.sender, assets[i]);
270         WithdrawalSchedule[] storage schedules = withdrawalSchedules[userToAsset
            ];

272         for (uint256 j = 0; j < clamiedWithdrawalSchedules.length; j++) {
273             if (clamiedWithdrawalSchedules[j] > 0 && clamiedWithdrawalSchedules[
                j] >= withdrawalscheduleCleanUp) {
274                 for (uint256 k = 0; k < schedules.length -
                    clamiedWithdrawalSchedules[j]; k++) {
275                     schedules[k] = schedules[k + clamiedWithdrawalSchedules[j]];

```

```

276         }
277
278         while (claimedWithdrawalSchedules[j] > 0) {
279             schedules.pop();
280             claimedWithdrawalSchedules[j]--;
281         }
282     }
283 }
284 }
285 }

```

**Remediation** Revisit the above mentioned functions to ensure all claimed schedules are accurately counted and properly removed from the schedules list.

### [H-5] Reserve ETH for New Validators in NodeDelegator::receive()

Target	Category	IMPACT	LIKELIHOOD	STATUS
NodeDelegator.sol	Business Logic	High	High	<a href="#">Addressed</a>

The NodeDelegator contract accepts the deposit of native token from the EigenpieStaking contract and initializes validators by depositing specified data onto the Beacon Chain. With the validators, they can participate in Native Restaking in EigenLayer.

However, while reviewing the implementation of the `receive()` routine, we notice that all the received native tokens are transferred to the `rewardDistributor` as rewards (line 48). As a result, there is no native token left for the NodeDelegator to setup validators.

#### NodeDelegator::receive()

```

40 receive() external payable {
41     // If a payment comes in from the delayed withdrawal router, assume it is
42     // from the pending unstaked withdrawal
43     // and subtract that amount from the pending amount
44     if (msg.sender == address(eigenPod.delayedWithdrawalRouter())) {
45         ...
46     }
47
48     address rewardDistributor = eigenpieConfig.getContract(EigenpieConstants.
49         EIGENPIE_REWADR_DISTRIBUTOR);
50     TransferHelper.safeTransferETH(rewardDistributor, msg.value);
51
52     emit RewardsForwarded(rewardDistributor, msg.value);
53 }

```

---

**Remediation** Properly reserve the received staking of native token within the contract to initialize validators.

## [H-6] Flawed Exchange Rate Calculation

Target	Category	IMPACT	LIKELIHOOD	STATUS
EigenpieStaking.sol EigenpieWithdrawManager.sol	Business Logic	High	Medium	<a href="#">Addressed</a>

Eigenpie calculates the exchange rate of mLRT by dividing the total amount of deposited LSTs by the totalSupply. The total amount of LSTs is obtained by calling `EigenpieStaking::getTotalAssetDeposits()`, which sums up the LSTs in the EigenpieStaking contract, NDCs, and EigenLayer but does not account for the LSTs in the EigenpieWithdrawManager contract. In particular, if the EigenpieWithdrawManager contract holds LSTs for users withdrawals, updating the exchange rate before the user's mLRT is burned would result in an incorrect value.

### EigenpieStaking::getAssetDistributionData

```
102 function getAssetDistributionData(address asset)
103     public
104     view
105     override
106     onlySupportedAsset(asset)
107     returns (uint256 assetLyingInDepositPool, uint256 assetLyingInNDCs, uint256
        assetStakedInEigenLayer)
108 {
109     assetLyingInDepositPool = TransferHelper.balanceOf(asset, address(this));
110
111     uint256 ndcsCount = nodeDelegatorQueue.length;
112     for (uint256 i; i < ndcsCount;) {
113         assetLyingInNDCs += TransferHelper.balanceOf(asset, nodeDelegatorQueue[i])
114         ;
115         assetStakedInEigenLayer += INodeDelegator(nodeDelegatorQueue[i]).
            getAssetBalance(asset);
116         unchecked {
117             ++i;
118         }
119     }
```

**Remediation** It's recommended to include the amount of LSTs held in the EigenpieWithdrawManager contract into the calculation within `EigenpieStaking::getTotalAssetDeposits()`.

## [M-1] Potential Risks Associated with Centralization

Target	Category	IMPACT	LIKELIHOOD	STATUS
Multiple Contracts	Security	Medium	Medium	Acknowledged

In the Eigenpie protocol, the existence of a series of privileged accounts introduces centralization risks, as they hold significant control and authority over critical operations governing the protocol. In the following, we show the representative function potentially affected by the privileges associated with the privileged accounts.

### MLRT::mint()/burnFrom()

```
67 /// @notice Mints EGETH when called by an authorized caller
68 /// @param to the account to mint to
69 /// @param amount the amount of EGETH to mint
70 function mint(address to, uint256 amount) external onlyRole(EigenpieConstants.
    MINTER_ROLE) whenNotPaused {
71     _mint(to, amount);
72 }

74 /// @notice Burns EGETH when called by an authorized caller
75 /// @param account the account to burn from
76 /// @param amount the amount of EGETH to burn
77 function burnFrom(address account, uint256 amount) external onlyRole(
    EigenpieConstants.BURNER_ROLE) whenNotPaused {
78     _burn(account, amount);
79 }
```

**Remediation** To mitigate the identified issue, it is recommended to introduce multi-sig mechanism to undertake the role of the privileged accounts. Moreover, it is advisable to implement timelocks to govern all modifications to the privileged operations.

**Response By Team** This issue has been confirmed by the team. The multi-sig mechanism will be used to mitigate this issue.

## [M-2] Revised assetTotalWithdrawAmt Check

Target	Category	IMPACT	LIKELIHOOD	STATUS
EigenpieWithdrawManager.sol	Coding Practices	Medium	Medium	<a href="#">Addressed</a>

The `EigenpieWithdrawManager::completeAssetWithdrawalFromEigenLayer()` function is responsible for completing asset withdrawals initiated from the EigenLayer. It aims to complete all withdrawals for the given assets in the specified epoch. Therefore, it checks if the total amount of assets to be

withdrawn in that epoch (i.e., `assetTotalWithdrawAmt`) can be fulfilled by the current input parameters. However, we notice that the current implementation incorrectly checks the `assetTotalWithdrawAmt`. Specifically, the current implementation requires withdrawing the `assetTotalWithdrawAmt` amount of assets from each NDC (line 228), while the expectation is to withdraw the `assetTotalWithdrawAmt` amount of assets from all NDCs combined.

#### EigenpieWithdrawManager::withdrawAssetsFromEigenLayer()

```

222 for (uint256 i = 0; i < nodeDelegators.length; i++) {
223     if (nodeToAssets[i].length != nodeToAmount[i].length) {
224         revert LengthMismatch();
225     }
226     for (uint256 j = 0; j < nodeToAssets[i].length; j++) {
227         bytes32 assetToEpoch = assetEpochKey(nodeToAssets[i][j], epochTime);
228         if (assetTotalWithdrawAmt[assetToEpoch] != nodeToAmount[i][j]) revert
            InvalidWithdrawAmt();
229     }
230 }

```

**Remediation** Revisit the implementation of `completeAssetWithdrawalFromEigenLayer()` function to improve the check for `assetTotalWithdrawAmt`.

## [L-1] Integration of Non-Standard ERC20 Tokens

Target	Category	IMPACT	LIKELIHOOD	STATUS
Multiple Contracts	Business Logic	Low	Low	<a href="#">Addressed</a>

Inside the `EigenpieStaking::depositAsset()` function, the statement of `if (!IERC20(asset).transferFrom(msg.sender, address(this), depositAmount)) {revert TokenTransferFailed();}` (line 69) is employed to transfer the user's asset into the `EigenpieStaking` contract. However, in the case of USDT-like token whose `transferFrom()` lacks a return value, it would lead to a revert. Given this, we recommend employing the widely-used `SafeERC20` library (which serves as a wrapper for ERC20 operations while accommodating a diverse range of non-standard ERC20 tokens) to address this case.

#### EigenpieStaking::depositAsset()

```

128 function depositAsset(
129     address asset,
130     uint256 depositAmount,
131     uint256 minRec,
132     address referral
133 )
134     external

```



```

135     whenNotPaused
136     nonReentrant
137     onlySupportedAsset(asset)
138 {
139     // checks
140     if (depositAmount == 0 || depositAmount < minAmountToDeposit) {
141         revert InvalidAmountToDeposit();
142     }

144     if (depositAmount > getAssetCurrentLimit(asset)) {
145         revert MaximumDepositLimitReached();
146     }

148     if (!IERC20(asset).transferFrom(msg.sender, address(this), depositAmount)) {
149         revert TokenTransferFailed();
150     }

152     // mint receipt
153     uint256 mintedAmount = _mintMLRT(asset, depositAmount);
154     if (mintedAmount < minRec) {
155         revert MinimumAmountToReceiveNotMet();
156     }

158     emit AssetDeposit(msg.sender, asset, depositAmount, referral);
159 }

```

**Remediation** Replace `transfer()/transferFrom()` with `safeTransfer()/safeTransferFrom()`.

## [L-2] Improved Logic of MLRTWallet::withdrawFromSwellStaking()

Target	Category	IMPACT	LIKELIHOOD	STATUS
MLRTWallet.sol	Business Logic	Low	High	<a href="#">Addressed</a>

The `withdrawFromSwellStaking()` function is used to withdraw the staked MLRT tokens from the `swellSimpleStaking` contract. Typically, only privileged accounts of the wallet can perform this withdrawal operation. However, if the user's collateral is insufficient, anyone can withdraw the excess MLRT tokens and burn them. While examining its implementation, we notice that if a malicious actor withdraws the excess MLRT tokens without burning them, he can repeatedly withdraw MLRT tokens until all tokens are exhausted, which clearly undermines the original design of the protocol.

### MLRTWallet::withdrawFromSwellStaking()

```

96 function withdrawFromSwellStaking (address mlrt, uint256 amount) external
    nonReentrant {

```

```

97     _checkValidWithdrawCondition(msg.sender, amount);

99     ISimpleStakingERC20 swellSimpleStaking = ISimpleStakingERC20(eigenpieConfig.
        getContract(EigenpieConstants.SWELL_SIMPLE_STAKING));
100    swellSimpleStaking.withdraw(IERC20(mlrt), amount, address(this));

102    emit WithdrawFromSwellStaking(msg.sender, mlrt, amount);
103 }

105 function _checkValidWithdrawCondition(address caller, uint256 amountToWithdraw)
    internal {
106     bool isClient = caller == client;
107     bool isClientOperator = allowedClientOperators[caller];
108     bool isManager = IAccessControl(address(eigenpieConfig)).hasRole(
        EigenpieConstants.MANAGER, caller);
109     // if client or eigenpie manager calling, then all good
110     if (isClient || isClientOperator || isManager) return;
111
112     (, uint256 mlrtLess) = eigenpieEnterprise.nativeRestakedLess(client);
113     if (amountToWithdraw > mlrtLess) revert PublicWithdrarTooMuch();
114 }

```

**Remediation** Improve the implementation of the `withdrawFromSwellStaking()/withdrawFromZicruit()` functions to prevent above-mentioned attack.

---

## 4 | Appendix

### 4.1 About AstraSec

AstraSec is a blockchain security company that serves to provide high-quality auditing services for blockchain-based protocols. With a team of blockchain specialists, AstraSec maintains a strong commitment to excellence and client satisfaction. The audit team members have extensive audit experience for various famous DeFi projects. AstraSec's comprehensive approach and deep blockchain understanding make it a trusted partner for the clients.

### 4.2 Disclaimer

The information provided in this audit report is for reference only and does not constitute any legal, financial, or investment advice. Any views, suggestions, or conclusions in the audit report are based on the limited information and conditions obtained during the audit process and may be subject to unknown risks and uncertainties. While we make every effort to ensure the accuracy and completeness of the audit report, we are not responsible for any errors or omissions in the report.

We recommend users to carefully consider the information in the audit report based on their own independent judgment and professional advice before making any decisions. We are not responsible for the consequences of the use of the audit report, including but not limited to any losses or damages resulting from reliance on the audit report.

This audit report is for reference only and should not be considered a substitute for legal documents or contracts.

---

## 4.3 Contact

<b>Name</b>	AstraSec Team
<b>Phone</b>	+86 176 2267 4194
<b>Email</b>	contact@astrasec.ai
<b>Twitter</b>	<a href="https://twitter.com/AstraSecAI">https://twitter.com/AstraSecAI</a>